
Subject: Protection with TT injections for most recent birth

Posted by [Rupon](#) on Mon, 27 Nov 2023 05:44:22 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi!

I am working with NFHS-5 (2019-21) data for India, and using KR file for the most recent live births (MIDX==1). I am trying to reproduce INDICATOR 42 in STATE FACT SHEET (MOTHERS WHOSE LAST BIRTH WAS PROTECTED WITH NEONATAL TETANUS).

As defined this indicator includes

- (a) mothers receiving 2 TT injection during pregnancy, OR
- (b) two or more injections (the last within 3 years of the last live birth), OR
- (c) three or more injections (the last within 5 years of the last birth), OR
- (d) four or more injections (the last within 10 years of the last live birth), OR
- (e) five or more injections at any time prior to the last birth.

I am currently using Stata package. Could anyone please provide DO FILE for it? Your support is appreciated.

Thank you.

Regards

Rupon

Subject: Re: Protection with TT injections for most recent birth

Posted by [Bridgette-DHS](#) on Tue, 28 Nov 2023 16:43:04 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

This indicator appears in the national final report, in table 8.7. The Stata code is included on GitHub in the generic chapter 9, for ANC indicators (https://github.com/DHSProgram/DHS-Indicators-Stata/blob/master/Chap09_RH/RH_ANC.do). I have slightly modified it and will paste it below I include the code for 2+ TT injections. This runs off the IR file but you could also apply it to the KR file, selecting bidx=1 and removing "_01" wherever it appears.

```
use "...IAIR7EFL.DTA" , clear
```

```
gen age=b19_01
```

```
gen period=60
```

```
* tetanus toxoid injections
```

```
recode m1_1 (0 1 8 9 . = 0 "No") (2/7 = 1 "Yes"), gen(rh_anc_toxinj)
```

```
replace rh_anc_toxinj = . if age>=period
```

```
label var rh_anc_toxinj "Received 2+ tetanus injections during last pregnancy"
```

```
* protection against neonatal tetanus
```

```

* temporary vars needed to compute the indicator
gen totet = 0
gen ttprotect = 0
replace totet = m1_1 if (m1_1>0 & m1_1<8)
replace totet = m1a_1 + totet if (m1a_1 > 0 & m1a_1 < 8)

* generating variable for date of last injection - will be 0 for women with at least 1 injection at last
pregnancy
gen lastinj = 9999
replace lastinj = 0 if (m1_1>0 & m1_1 <8)
gen int ageyr = (age)/12

* years ago of last shot - (age at of child), yields some negatives
replace lastinj = (m1d_1 - ageyr) if m1d_1 <20 & (m1_1==0 | (m1_1>7 & m1_1<9996))

* generate summary variable for protection against neonatal tetanus
replace ttprotect = 1 if tet2lastp ==1

* at least 2 shots in last 3 years
replace ttprotect = 1 if totet>=2 & lastinj<=2

* at least 3 shots in last 5 years
replace ttprotect = 1 if totet>=3 & lastinj<=5

* at least 4 shots in last 10 years
replace ttprotect = 1 if totet>=4 & lastinj<=9

* at least 2 shots in lifetime
replace ttprotect = 1 if totet>=5
label variable ttprotect "Full neonatal tetanus Protection"

gen rh_anc_neotet = ttprotect
replace rh_anc_neotet = . if bidx_01!=1 | age>=period
label variable rh_anc_neotet "Protected against neonatal tetanus"

tab rh_anc_toxinj [iweight=v005/1000000]
tab rh_anc_neotet [iweight=v005/1000000]

```

Subject: Re: Protection with TT injections for most recent birth
 Posted by [Rupon](#) on Wed, 29 Nov 2023 08:25:09 GMT
[View Forum Message](#) <> [Reply to Message](#)

Thank you much for your reply with detail codes.

Could you please elaborate on why m1d_1<20?

Subject: Re: Protection with TT injections for most recent birth
Posted by [Bridgette-DHS](#) on Wed, 29 Nov 2023 13:52:12 GMT
[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

This variable is the number of years since the most recent tetanus injection. If you look at the distribution, it goes out to 40 years, but 99.4% of the responses are in the past 20 years. I believe 20 years is an arbitrary boundary for saying that any larger numbers are probably due to recall error.

I just copied (with some minor editing) the GitHub construction, which is based on the CSPro construction. I don't know who decided on a boundary of 20 but it gives a match to the report.
